
Abstract: We discuss algorithmic optimization techniques for floating point DSPs, which are independent of architectural differences and provide tools for writing the fastest floating point DSP algorithms for any required precision.
 We show algorithms for polynomial evaluation used generally as an approximation tool. These algorithms are two to four times faster than the straightforward implementation of Horner's rule.
We present the fastest known algorithms for 1/x, 

, sin(x), atan(x/y) at various precision, and the nonlinear median-of-3 filter.
 Examples are in the TMS320C3x assembly language. A simple algorithm is also presented for finding approximation polynomials to any precision used at the above algorithms.








Introduction

There are many applications where signal processing involves a large number of evaluations of mathematical functions. Faster algorithms let us use lower DSP clock frequencies, which reduces costs, power (heat) and size.

Only the most frequently used math functions are discussed here, but the design- and optimization-techniques involved are more generally applicable. The algorithms are presented at different accuracy and are the fastest than any others we have seen for the TMS320C3x DSP family. Other DSPs have different addressing modes, instruction set, so the fastest algorithms might differ a little, but the optimization principles are similar.

Some applications could benefit from look-up tables, although a few of the presented routines are even faster and require very little memory (e.g. 2 instructions for 1/x with a maximum relative error of 1/8, 3 instructions for scaled 

, etc). Look-up tables are better suited for fixed-point DSPs. They might be the best solution if

· the range of the arguments is not large relative to the available memory
· the required accuracy (resolution) is low.
The functions as presented are fully tested and directly callable from C programs.



 Different applications need different levels of error checking and exception handling. These have to be added to the examples shown in this paper.
DSP architecture
There are many different types of floating point DSPs. The most common features, which distinguish them from other kinds of microprocessors are the following:
· independent floating point adder(s) and multiplier(s) allowing concurrent operations  
· independent ALUs for fix point addition, address calculation and auto-increment indexing
· zero overhead loop – delayed branch

· instruction pipeline, cache

· conditionally executed instructions
· fast memory interface
· large internal data storage.
Optimization Techniques
A typical DSP program contains several simple loops each processing the whole data set. Since the code of small loops fits in the internal cache, the data fetch and loop operations require no extra processing time. Loop unrolling only helps to speed up the shortest loops. Below we will see examples of the following optimization techniques:
· Choosing the best algorithm

· Taking advantage of built in data conversion instructions

· Interleaving

· Conditionally executing instructions instead of Branch
· Addressing jump tables by combined conditions
An Example DSP
We take as an example the Texas Instruments TMS320C3x family of DSPs. The cheapest among them, the C32, costs under $10 in small quantities (mid '99). Their full description and the assembly language conventions can be found in [3] and [4].
Architecture

· Arithmetic registers for real and integer numbers (R0…R7) and auxiliary registers (AR0…AR7) for addresses and integer data and the usual set of special registers (status-, index-, loop control-registers).

· Instructions fetched from external RAM can be cached.

· 2 blocks of internal (on-chip) RAM for data and program storage.

· Up to 3 memory accesses per instruction cycle: Instruction fetch from the cache + Data access from internal RAM-block 1 and 2 when accessed via register indirection. Instead of RAM, registers can be used, but with certain restrictions about the allowed combinations of RAM and register parameters.

· Parallel instructions, e.g.: multiply‑add, load‑store executed in one instruction cycle.

· Pointers can be incremented, decremented in parallel with the instruction executions.

· 3 instructions are pipelined. Branches flush the pipeline, and take 4 cycles to execute.

· Delayed branch instructions perform the next 3 instructions, before the jump takes place. They effectively consume only 1 DSP cycle to execute.

Representation of Numbers

Integers (fix-point numbers) are stored in registers or memory as 32 bit 2's complement binary entities.

Floating-point numbers are stored in the arithmetic registers as 40 bit binary entities. The most significant 8 bits constitute the exponent e in 2's complement binary form. The least significant 32 bits encode the mantissa m in the following form: the most significant bit is the sign s, the remaining 31 bits form a fix-point number f assuming a binary point in front of it (0 ≤ f < 1): eeeeeeee s fffffffffffffffffffffffffffffff
· If s = 0, the number is positive and m = f + 1.

· If s = 1, the number is negative and m = f ( 2.

The resulting floating-point number is 

, where 1 ≤ m < 2
 or –2 ≤ m < –1
. 
They are stored in memory as the most significant 32 bits of the 40 in the arithmetic registers. The 8 least significant bits are ignored.

C3x C-compiler register argument passing conventions

The details can be found in [5].

Registers R2, R3 are used for the 1st two float arguments

· AR2, R2, R3, RC, RS, RE are used for the remaining scalar arguments

· Further parameters are pushed onto the stack right to left

· float or int return-value is given in R0

· Pointer return-value is given in AR0

· The content of registers: AR3-AR7, SP, DP; int R4, R5, float R6, R7 must be preserved

· The return address is pushed onto the stack

Polynomial evaluation

Polynomials provide a flexible approximation tool for any smooth function. We use them everywhere, therefore the evaluation algorithms play a crucial role at calculation of almost every math function.
Inner product

The most straightforward way to evaluate a real polynomial

at a given real x is to calculate and store the powers X = (

 and compute the inner product with the coefficient vector A = (

. The first step requires n+const DSP instructions (n multiply-stores plus initialization) with some small constant const. The inner product calculation also requires n+const DSP instructions (n+1 multiply-adds plus initialization). All together about 2n steps, which is not optimal. However, there are situations when this method is not bad: if many polynomials have to be evaluated at the same x. The vector X is calculated once and used in all the polynomial evaluations.

Horner's rule

Horner's rule to evaluate a polynomial is based on the following formula

,

which requires n multiplications and n additions. These numbers are optimal if neither x nor A is known in advance, that is, the polynomial cannot be pre-processed beforehand.

One might think, that given a DSP the multiplications and additions could be performed in parallel, speeding up the process twofold. Unfortunately, this is not the case. Horner's rule is inherently serial. We need the result of the last multiplication before we can start the next addition, and its result is needed for the next multiplication.

Interleaving

When performing Horner's rule, either the multiplier or the adder is idle. But it also means, we can do 2 polynomial evaluations in parallel. If a process performs an addition, the other multiplies and vice versa. Thus 2 polynomial evaluations require 2n+const DSP steps. It is an effective twofold improvement, if we arrange the computation such that the polynomial evaluations occur in pairs. In typical DSP applications this is not difficult.
; Horner2(float xa, float xb, float *A, float *B, int Len, float *R)

;            R2       R3         AR2        RC      RS > 2    RE


; A[0]xa^(Len-1) + A[1]xa^(Len-2) + ... + A[Len-1]  -> R[0]

; B[0]xb^(Len-1) + B[1]xb^(Len-2) + ... + B[Len-1]  -> R[1]

; 2 cycles/degree; Optimum speed if A and B in the same internal RAM block

Horner2Data  .word      Horner2Data+1   ; <- other RAM block than A and B

             .word      0               ; *AR0  to store xa

             .word      0               ; *+AR0 to store xb

_Horner2      PUSH      RE              ; *R

              LDI       @Horner2Data,AR0; constant!!!

              LDI       RC,AR1          ; *B (constant!)

              MPYF      *AR2++,R2,R0    ; pa = A[1]xa

           || STF       R2,*AR0         ; store xa

              MPYF      *AR1++,R3,R1    ; pb = B[1]xb

           || STF       R3,*+AR0        ; store xb

              ADDF      *AR2++,R0,R2    ; pa += A[2] (has no || pair)

              LDI       RS,RC

              SUBI      3,RC

              RPTB      HorLoop

              MPYF      R2,*AR0,R0      ; pa *= xa

           || ADDF      *AR1++,R1,R3    ; pb += B[i]

HorLoop       MPYF      R3,*+AR0,R1     ; pb *= xb

           || ADDF      *AR2++,R0,R2    ; pa += A[i]

              ADDF      *AR1++,R1,R3    ; pb += B[Len] (has no || pair)

              POP       AR1             ; get pointer to output

              STF       R2,*AR1         ; store pa(xa)

           || STF       R3,*+AR1        ; store pb(xb)

              RETS
Product of 2 factors

In general, if a polynomial is used to approximate a known function, its coefficients can be determined in advance and the polynomial can be brought into a form which allows faster evaluation
. Real polynomials can be factored in two equal-degree (within 1) polynomial factors. The double-Horner's rule shown above can then calculate both the polynomial factors in n+const steps. Multiplying the results together yields a method for polynomial evaluation with n+const DSP instructions.

Product of linear and quadratic factors

Real polynomials can be factored into 1st and 2nd order terms. With multiply‑add instructions, the product of k first order terms is easily calculated in k+const steps and the product of m second order terms in 2m+const steps. A total of n+const DSP instructions for the whole polynomial. The product of the terms 

 is calculated the following way:

1. 

 are calculated in parallel (

 should be pre-calculated )

2. sum them in parallel with the cumulative multiplication to the previous quadratic terms:

; float qdprod(float x, float *A, float *B, int Len) = product of (x^2 + A[i] x + B[i])
;                  R2      AR2       R3 > 0    RC



; Len+1 entries are read from A and B, Len entries used, to save code

; 2 cycles/quadratic factor; Optimum speed if A and B in the same internal RAM block

qdprodData   .word      qdprodData+1    ; <- other RAM block than A and B

             .word      0               ; *AR0  to store x

             .word      0               ; *+AR0 to store x^2

_qdprod       LDI       @qdprodData,AR0 ; constant!!!

              LDI       R3,AR1          ; *B (constant!)

              MPYF      R2,R2,R3        ; x^2

              STF       R3,*+AR0        ; store x^2 to *+AR0

           || STF       R2,*AR0         ; store  x  to *AR0

              LDF       1.0,R0          ; initial product

              MPYF      *AR2++,R2,R1    ;   A[1]x

           || ADDF      *AR1++,R3,R2    ; x^2 + B[1]

              SUBI      1,RC

              RPTB      qdLoop

              MPYF      *AR2++,*AR0,R1  ; A[i+1]x             |

           || ADDF      R2,R1,R3        ; x^2 + A[i]x + B[i]  |

qdLoop        MPYF      R3,R0           ; product             |

           || ADDF      *AR1++,*+AR0,R2 ; x^2 + B[i+1]        |

              RETS

Eve's method

Eve's method to evaluate a polynomial is the following

where 

 with appropriate parameters. Eve proved the surprising fact (see [1]), that every polynomial can be written in this form. It is important for computers where the multiplications are much more expensive than additions, because the number of multiplications is n/2+const (while the number of additions is still n+const), and thus it is significantly faster than the Horner's rule.

Since there are n additions, this method is no faster on a DSP than the previously described methods. It is not slower, either, therefore it presents another alternative for polynomial evaluations. Eve's method is most useful for polynomial approximations that have been used and tested on traditional computers. They are often in Eve's form already.

; float EVEpol(float x, float x0, float *A, float *B, int Len)

;                  R2       R3      AR2        RC > 0    RS


; = (...( (x0*(x+A[1]) + B[1])) * (x+A[2]) + B[1]) )...) * (x+A[Len]) + B[Len]

; Len+1 entries are read from A, Len entries used, to save code

; 2 cycles/EVE step; Optimum speed if A and B in the same internal RAM block

EVEpolData   .word      EVEpolData+1    ; <- other RAM block than A and B

             .word      0               ; *AR0  to store x

_EVEpol       LDI       @EVEpolData,AR0 ; constant!!!

              LDI       RC,AR1          ; *B (constant!)

              LDF       R3,R0           ; initial pol -> R0

              ADDF      *AR2++,R2       ; x + A[1]    -> R2

           || STF       R2,*AR0         ; store  x  to *AR0

              LDI       RS,RC

              SUBI      1,RC

              RPTB      EVLoop

              MPYF      R2,R0           ; pol *= (x + A[i])   |

           || ADDF      *AR2++,*AR0,R2  ; x + A[i+1] -> R2    |

EVLoop        ADDF      *AR1++,R0       ; + B[i]              |

              RETS
If our DSP had two adders, Eve's method could be implemented with n/2 steps, keeping the 2 adders and the multiplier constantly busy providing a four fold speed-up over Horner’s rule.
Reciprocal

The fastest way to calculate the reciprocal 1/a is based on Newton's method. 1/a is the solution of the equation ax = 1. Starting from an initial 

. If the relative accuracy of 

, i.e., the number of accurate bits doubles at each iteration. Furthermore, the error is always negative, therefore scaling the result by a constant gains an additional accurate bit. This scaling usually need not be done explicitly because its effect can be considered at the following calculations.

There are similar iterative formulae, which increase the number of accurate bits by a larger factor but by themselves they are less economical since they require more calculations relative to the accuracy gain. E.g.: 

 needs 5 DSP instructions to triple the number of accurate bits compared to 3 instructions of the previous formula to double it. Interleaving 2 reciprocal calculations as discussed at the Horner's rule could change the situation, but the increase of complexity seldom justifies higher order iterative methods.

Finding a good initial estimate is crucial. Here fiddling with the bits of the floating point numbers helps. 

, so we need to negate the exponent. Flipping all its bits almost does it: we get 

. We don't want to change the sign bit, of course, but how about flipping the mantissa bits, too? With the notation of the Number Representation section f gets replaced by 1–f. That is, the bit flipping gives:

· If x > 0: 
[image: image54.wmf]¢

=

-

m

m

3


,   
[image: image56.wmf]x

m

x

m

e

e

=

¢

=

×

-

®

-

2

2

3

2


.

· If x < 0: 

This bit flipping gives a good approximation, because 

. The error is at most 1/8, that is, we have at least 3 accurate bits to start the Newton iteration with.

; float reci4(float *x) 
= 1/(*x) within +/- 5.88% relative error (>4 bit)

; if *x = 0, result = 3.2e+38 around max float = 3.4e+38

; 3 DSP instructions + Initialization and CALL/RETURN

reci4data    .word      reci4data+2     ; allocate in the best RAM block

             .word      0FF7FFFFFh      ; *-AR0

             .word      0               ; *AR0 Middle of constant block

             .float     0.9412          ; *+AR0

_reci4:       LDI       @reci4data,AR0  ; constant!!!

              XOR       *AR2,*-AR0,R0   ; flip all but sign bit <= work done

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      *AR0,*+AR0,R0   ; make error symmetric

              RETS

; float reci9(float *x)
 = 1/(*x) within +/- 0.173 % relative error (>9 bit)

; if *x = 0, result = 3.0194955e+38 around max float = 3.4e+38

; 6(5) DSP instructions + Initialization and CALL/RETURN

reci9data    .word      reci9data+2     ; allocate in the best RAM block

             .word      0FF7FFFFFh      ; *-AR0

             .word      0               ; *AR0 Middle of constant block

             .float     0.88735         ; *+AR0

_reci9:       LDI       @reci9data,AR0  ; constant!!!

              XOR       *AR2,*-AR0,R0   ; flip all but sign bit

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      *AR2,*AR0,R1    ; xy

              SUBRF     2.125,R1        ; 2.125-xy [2.125 = 2/0.9412: symmetric error]

              MPYF      R1,*AR0,R0      ; y*(2.125-xy)

              MPYF      *+AR0,R0        ; error corrctn - outside or || next SUBRF

              RETS

; float reci12(float *x)
 = 1/(*x) within +/- 0.02 % relative error (12.3 bit)

; if *x = 0, result = max float = 3.4e+38

; 8(7) DSP instructions + Initialization and CALL/RETURN (good for interleaving)

reci12data   .word      reci12data+2    ; allocate in the best RAM block

             .word      0FF7FFFFFh      ; *-AR0

             .word      0               ; *AR0 Middle of constant block

             .float     3.18742         ; *+AR0

             .float     0.833644        ; *+AR0(2)

_reci12:      LDI       @reci12data,AR0 ; constant!!!

              XOR       *AR2,*-AR0,R0   ; flip all but sign bit

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      *AR2,*AR0,R1    ; x*yd

              SUBF      *+AR0,R1,R0     ; x yd-3.2

              MPYF      R1,R0           ; x yd(x yd-3.2)

              ADDF      3.38655,R0      ; 3.4+x yd(x yd-3.2)

              MPYF      *AR0,R0         ; yd[3.4+x yd(x yd-3.2)]

              MPYF      *+AR0(2),R0     ; symmetrize - outside or || next ADDF

              RETS

; float reci19(float *x)
 = 1/(*x) within +/- 1.7e-6 relative error (>19 bit)

; if *x = 0, result = 2.7e+38, close to max float = 3.4e+38

; 9(8) DSP instructions + Initialization and CALL/RETURN

reci19data   .word      reci19data+2    ; allocate in the best RAM block

             .word      0FF7FFFFFh      ; *-AR0

             .word      0               ; *AR0 Middle of constant block

             .float     0.78738832      ; *+AR0

_reci19:      LDI       @reci19data,AR0 ; constant!!!

              XOR       *AR2,*-AR0,R0   ; flip all but sign bit

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      *AR2,*AR0,R1    ; xy

              SUBRF     2.125,R1        ; 2.125-xy [2.125 = 2/0.9412: symmetric error]

              MPYF      R1,*AR0,R0      ; y*(2.125-xy)

              MPYF      *AR2,R0,R1      ; xy

              SUBRF     2.2539,R1       ; 2.2539-xy [2/0.88735: symmetric error]

              MPYF      R1,R0           ; y*(2.2539-xy)

              MPYF      *+AR0,R0        ; symmetrize - outside or || next SUBRF
              RETS

Square root

Instead of calculating 

 directly, we approximate 

The fastest method is also based on Newton's method. 

. If the relative accuracy of 

, that is, the number of accurate bits nearly doubles at each iteration. Furthermore, the error is negative (if it is small), therefore scaling the result by a constant to get equal positive and negative peak error gains a further accurate bit. This scaling usually need not be done explicitly, its effect can be considered at subsequent calculations.

There are similar iterative formulae, which increase the number of accurate bits by a larger factor, but alone they are less economical requiring more calculations relative to the accuracy gain. E.g.: 

 needs 7 DSP instructions to triple the number of the accurate bits compared to 4 instructions of the previous formula to double it. Interleaving 2 reciprocal calculations as discussed at the Horner's rule could change the situation, but the increase of complexity seldom justifies higher order iterative methods.

Finding a good initial estimate is again crucial, and manipulating the bits of the exponent and mantissa still helps. 

, so we need to negate and halve the exponent. Flipping all its bits and shift them right almost does it: we get an exponent 

, so it helps flipping and shifting the mantissa bits, too. The sign bit must be 0, since negative numbers are not allowed for the square root. The least significant bit of the exponent is shifted to the sign bit. It has to be cleared. Corresponding to its value we can act differently to improve the approximation.

 The result of the bit manipulation starting from x:
 If e was 0 (1 ≤ x < 2): 

To improve the approximation we apply a correction factor of 0.9634 for the first case and 0.9634 /

= 0.6813 for the second case. The error is at most 3.6% (in other words, we have almost 5 accurate bits to start the Newton iteration with).

SqMask        .word     000800000h
temp          .word     0h
tempPTR       .word     temp
; float rsqt5(float *x)
 = 1/sqrt(x) within +/- 3.6% relative error (4.8 bit)

; if *x = 0, result = 1.777e+19

; if *x < 0, result is 1/sqrt(-x) with larger error

; 8 DSP instructions + Initialization and CALL/RETURN

_rsqt5:       LDI       @tempPTR,AR0    ; constant!!!

              NOT       *AR2,R0         ; flip all bits: 1/x

              ASH       -1,R0           ; exponent halved

              TSTB      @SqMask,R0      ; test last exponent bit

              LDFZ      0.6813,R3       ; centralize error

              LDFNZ     0.9634,R3       ; sqrt(2) correction

              ANDN      @SqMask,R0      ; Fix sign to positive

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      *AR0,R3,R0      ; error correction

              RETS

; float rsqt10(float *x)
 = 1/sqrt(x) within +/- 0.1% relative error (10 bit)

; if *x = 0, result = 2.67e+19

; 12 DSP instructions + Initialization and CALL/RETURN

_rsqt10:      LDI       @tempPTR,AR0    ; constant!!!

              NOT       *AR2,R0         ; flip all bits: 1/x

              ASH       -1,R0           ; exponent halved

              TSTB      @SqMask,R0      ; test last exponent bit

              LDFZ      0.54057,R3      ; centralize error

              LDFNZ     0.76479,R3      ; sqrt(2) correction

              ANDN      @SqMask,R0      ; Fix sign to positive

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      *AR0,R3,R0      ; approx/sqrt(2)

              MPYF      R0,R0,R3        ; y^2

              MPYF      *AR2,R3         ; x y^2

              SUBRF     1.89114,R3      ; 1.5 d^2 - x y^2

              MPYF      R3,R0           ; y (1.5 d^2 - x y^2)

              RETS

; float rsqt20(float *x)
 = 1/sqrt(x) within +/- 1e-6 relative error (20 bit)

; if *x = 0, result = 4.00e+19

; 16 DSP instructions + Initialization and CALL/RETURN

rsqt20       .float     1.8898829       ; not immediate param, >20 bit accurate

_rsqt20:      LDI       @tempPTR,AR0    ; constant!!!

              NOT       *AR2,R0         ; flip all bits: 1/x

              ASH       -1,R0           ; exponent halved

              TSTB      @SqMask,R0      ; test last exponent bit

              LDFZ      0.50082,R3      ; centralize error

              LDFNZ     0.70820,R3      ; sqrt(2) correction

              ANDN      @SqMask,R0      ; Fix sign to positive

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      *AR0,R3,R0      ; approx/sqrt(2)

              MPYF      R0,R0,R3        ; y^2

              MPYF      *AR2,R3         ; x y^2

              SUBRF     1.62113,R3      ; 1.5 d^2 - x y^2

              MPYF      R3,R0           ; y (1.5 d^2 - x y^2)

              MPYF      R0,R0,R3        ; y^2

              MPYF      *AR2,R3         ; x y^2

              SUBRF     @rsqt20,R3      ; 1.5 D^2 - x y^2

              MPYF      R3,R0           ; y (1.5 D^2- x y^2)

              RETS

There is another way to clear the sign-bit: shift right the mantissa only. The result of this bit manipulation from x: If e was 0 (1 ≤ x < 2): 

This method is 2 or 3 instructions faster, but less accurate (11%). It serves also well as a first step in the Newton iteration.
; float rsqt3(float *x) = 1/sqrt(x) within +/- 0.11 relative error (3.2 bit)

; 5 DSP instructions + 1 scaling + Initialization and CALL/RETURN

_rsqt3:

              LDI       @tempPTR,AR0    ; constant!!!

              NOT       *AR2,R0         ; flip all bits: 1/x

              ASH       -1,R0           ; exponent halved

              STI       R0,*AR0         ; write 32 bits to RAM

              LDF       *AR0,R0         ; mantissa in integer part of R0

              LSH       -1,R0           ; clear sign bit

              MPYF      0.8889,R0       ; scaling - outside

              RETS
; float rsqt7(float *x) = 1/sqrt(x) within +/- 0.0093 relative error (6.75 bit)

; 9 DSP instructions + 1 scaling + Initialization and CALL/RETURN

_rsqt7:

              LDI       @tempPTR,AR0    ; constant!!!

              NOT       *AR2,R0         ; flip all bits: 1/x

              ASH       -1,R0           ; exponent halved

              STI       R0,*AR0         ; write 32 bits to RAM

              LDF       *AR0,R0         ; mantissa in integer part of R0

              LSH       -1,R0           ; clear sign bit

              MPYF      R0,R0,R3        ; y^2

              MPYF      *AR2,R3         ; x y^2

              SUBRF     3.812788,R3     ; 1.5 d^2 - x y^2

              MPYF      R3,R0           ; y (1.5 d^2 - x y^2)

              MPYF      0.3523,R0       ; scaling - outside

              RETS

Adding another Newton iteration step increases the accuracy to 14 bits:

and so on, each additional Newton step (4 DSP cycles) doubles the number of accurate bits.

Logarithm

The DSP’s fix- and floating-point conversion instructions can help us to calculate the binary (base 2) logarithm. There is a simple conversion to any other positive base (b ( 1): 

. The most often used cases:

The simplest logarithm calculation method is based on the algorithm described in [6]. The exponent is the integer part of the binary logarithm. The fractional part has to be calculated from the mantissa m = 1+ f. If we just append f to the exponent, it is equivalent to the approximation 

), which is exact at the endpoints and has a maximum discrepancy of 0.086 in between.


; float logb4(float *x)
 = log2(x) within absolute error +/- 0.043

; dB calculation *= 3.0103000; error +/- 0.13 dB; 
if *x = 0, result = -127.9996


; 6(3) DSP instructions + Initialization and CALL/RETURN

pow_24       .float     5.96046447754e-8; 2^-24

_logb4:       LDF       *AR2,R2         ; x into a register

              ASH       1,R2            ; move mantissa over its sign

              STF       R2,@temp        ; store (exp,mantissa)

              FLOAT     @temp,R0        ; float(float) = log  <== work done

              MPYF      @pow_24,R0      ; scale        - can be outside

              ADDF      0.043,R0        ; center error - can be outside

              RETS

The fractional part can be more accurately calculated from the mantissa m = 1+ f. Any precision can be achieved by an appropriate polynomial approximation 

). Here we present a 3rd order approximation:

; float logb11(float *x)
 = log2(x) with absolute error +/- 0.000677 (10.5 bit)

; dB calculation *= 3.0103000; error +/- 0.002 dB; 
if *x = 0, result = -127.9996

; 12 DSP instructions + Initialization and CALL/RETURN

logb11       .float     1.661435

_logb11:      LDF       *AR2,R1          ; get x

              LDE       1.0,R1           ; clear exponent => mantissa

              LDF       R1,R0            ; R0 = m (1<= m < 2)

              MPYF      0.16039,R0       ;  0.16m

              SUBF      1.06199,R0       ;  0.16m-1.06

              MPYF      R1,R0            ; (0.16m-1.06)m

              ADDF      2.56343,R0       ; (0.16m-1.06)m+2.56

              MPYF      R1,R0            ;((0.16m-1.06)m+2.56)m

              SUBF      @logb11,R0       ;((0.16m-1.06)m+2.56)m-1.66

                        ; if needed higher order polynomial approximation here

              FLOAT     *AR2,R1          ; exp*2^24+(mantissa-1)*2^23

              MPYF      @pow_24,R1       ; scale by 2^-24

              ADDF      R1,R0            ; + exp

              RETS

Exponential

The DSP’s fix- and floating-point conversion instructions can help us to calculate powers of 2. There is a simple conversion to any other positive base power 

The simplest exponential algorithm converts x to integer. If x was first scaled by 224
, the bits of the integer form which fall into the exponent field of the floating point template are really the exponent of 2x
. We are almost done by taking this number as float. Only the sign bit must be cleared (by a logical right shift of the mantissa bits).

2x = 2(x(·2{x}
, where {x} denotes the fractional part of x. After the shift the mantissa is m = 1+ f  = 1+{x} which is a good approximation of 2{x}
 if  1<{x}<2.

; float expb5(float *x)
 = 2^x within relative error +/- 3% (5 bit accurate)

; 6(5) DSP instructions + Initialization and CALL/RETURN

expb5data    .word      expb5data+2     ; allocate in the best RAM block

             .float     16777216.0      ; *-AR0  2^24

             .word      0               ; *AR0 Middle of constant block

             .float     0.97018         ; *+AR0

_expb5        LDI       @expb5data,AR0  ; constant!!!

              MPYF      *AR2,*-AR0,R0   ; scale x

              FIX       R0              ; 2^x <= work done

              STI       R0,*AR0         ; store (exp,mantissa)

              LDF       *AR0,R0         ; get as float

              LSH       -1,R0           ; clear sign bit

              MPYF      *+AR0,R0        ; centralize error - can be outside

              RETS

If we want a better approximation of 2{x}
, take the mantissa m = 1+{x} by setting the exponent to 0, and perform a polynomial approximation of 2m–1 ( p(m)
 (1 < m < 2). We have chosen to improve the previously described approximation 2m–1 ( m
 by a correction factor, that is, the approximation polynomial is in the special form of m·p(m)
.

; float expb8(float *x)
 = 2^x within elative error +/- 0.35% (8 bit accurate)

; 13 DSP instructions + Initialization and CALL/RETURN

expb8data    .word      expb8data+2     ; allocate in the best RAM block

             .float     16777216.0      ; *-AR0  2^24

             .word      0               ; *AR0 Middle of constant block

             .float     1.456753        ; *+AR0

_expb8        LDI       @expb8data,AR0  ; constant!!!

              MPYF      *AR2,*-AR0,R0   ; scale x

              FIX       R0              ; 2^x <= work done

              STI       R0,*AR0         ; store (exp,mantissa)

              LDF       *AR0,R0         ; get as float

              LSH       -1,R0           ; clear sign bit

              LDF       R0,R2           ; first approximation

              LDE       1.0,R2          ; mantissa (1 <= m < 2)

              LDF       R2,R1           ; m

              MPYF      0.2336,R2

              SUBF      0.69376,R2      ; polynomial approx

              MPYF      R1,R2

              ADDF      *+AR0,R2        ; correction factor

              MPYF      R2,R0

              RETS

Polynomials of relative high degree are necessary for better approximations. 18-bit accuracy requires a 6th degree polynomial. (Interleaving saves 6 DSP cycles.)

; float expb18(float *x)
 = 2^x within a relative error +/- 3.3e-6 (18.2 bit accurate)

; 21(15 interleaved) DSP instructions + Initialization and CALL/RETURN

expb18data   .word      expb18data+2    ; allocate in the best RAM block

             .float     16777216.0      ; *-AR0  2^24

             .word      0               ; *AR0 Middle of constant block

             .float     0.03798636071274; *+AR0(1)

             .float    -0.39546710599940; *+AR0(2)

             .float     1.74975989620481; *+AR0(3)

             .float    -4.25090607856171; *+AR0(4)

             .float     6.17968884264571; *+AR0(5)

             .float    -5.16277631223648; *+AR0(6)

             .float     2.84171108945846; *+AR0(7)

_expb18       LDI       @expb18data,AR0 ; constant!!!

              MPYF      *AR2,*-AR0,R0   ; scale x

              FIX       R0              ; 2^x <= work done

              STI       R0,*AR0         ; store (exp,mantissa)

              LDF       *AR0,R0         ; get as float

              LSH       -1,R0           ; clear sign bit

              LDF       R0,R2           ; first approximation

              LDE       1.0,R2          ; mantissa (1 <= m < 2)

              LDF       R2,R1           ; m

              MPYF      *+AR0(1),R2     ; polynomial approximation

              ADDF      *+AR0(2),R2

              MPYF      R1,R2

              ADDF      *+AR0(3),R2

              MPYF      R1,R2

              ADDF      *+AR0(4),R2

              MPYF      R1,R2

              ADDF      *+AR0(5),R2

              MPYF      R1,R2

              ADDF      *+AR0(6),R2

              MPYF      R1,R2

              ADDF      *+AR0(7),R2

              MPYF      R2,R0           ; adjust 1st approximation

              RETS

Sine of small angles

For small angles (between –( /4
 and ( /4
), sin(x) can be surprisingly well appro​ximated by a low degree polynomial. Since sin(x) is an odd function, the best appro​xi​mation polynomial consists of only odd powers of x, that is, it is in the form x·p(x2)
.

; float sin20(float x)
 = sin(x), Abs error +/- 6e-7 (20.7 bit) if |x| < pi/4

; 6 DSP instructions + Initialization and CALL/RETURN

nsin20data   .word      nsin20data+1    ; -sin

             .float    -0.00812161743260; *AR0

             .float     0.16660166350892; *+AR0

             .float    -0.99999500208779; *+AR0(2)

sin20data    .word      sin20data+1     ; allocate in the best RAM block

             .float     0.00812161743260; *AR0

             .float    -0.16660166350892; *+AR0

             .float     0.99999500208779; *+AR0(2)

_sin20        LDI       @sin20data,AR0  ; constant!!!

              MPYF      R2,R2,R3        ; x^2

              MPYF      *AR0,R3,R0

              ADDF      *+AR0,R0

              MPYF      R3,R0

              ADDF      *+AR0(2),R0

              MPYF      R2,R0

              RETS

Cosine of small angles

For small angles (between –( /4
 and ( /4
), cos(x) can also be very well approximated by a relative low degree polynomial. Since cos(x) is an even function, the best approximation polynomial consists of only even powers of x. It is in the form p(x2)
.

; float cos16(float x)
 = cos(x), Abs error +/- 1e-5 (16.6 bit) if |x| < pi/4

; 5 DSP instructions + Initialization and CALL/RETURN

; The next, better approximation (+2 instr) is 25 bits accurate (+/- 2.76e-8):

;-0.001358595001140x^6+0.041655030391947x^4
-0.499998567517263x^2+0.999999972385343
ncos16data   .word      ncos16data+1    ; -cos

             .float    -0.0403990266    ; *AR0

             .float     0.4997084539    ; *+AR0

             .float    -0.9999900572    ; *+AR0(2)

cos16data    .word      cos16data+1     ; allocate in the best RAM block

             .float     0.0403990266    ; *AR0

             .float    -0.4997084539    ; *+AR0

             .float     0.9999900572    ; *+AR0(2)

_cos16        LDI       @cos16data,AR0  ; constant!!!

              MPYF      R2,R2,R3        ; x^2

              MPYF      *AR0,R3,R0

              ADDF      *+AR0,R0

              MPYF      R3,R0

              ADDF      *+AR0(2),R0

              RETS

Sine of Any Angle

For the general case we have to reduce the argument x to a finite interval. The identities 

tell us which of the above reduction formulae applies (all trigonometric functions are periodic with a period 2().

; float sine(float x)
 = sin(x), Absolute error at most +/- 1e-5 (16.6 bit)

; 16 or 17  DSP instructions + Initialization and CALL/RETURN

             .float     0.63661977236758; *-AR0   2/pi

             .word      SineJump-2      ; *AR0 Middle of constant block

             .float     1.5707963267949 ; *+AR0   pi/2

SineJump     .word      quart0          ; 0

             .word      quart1          ; 1

             .word      quart2          ; 2

             .word      quart3          ; 3

_sine         LDI       @SineJump-2,AR0 ; constant!!!

              MPYF      *-AR0,R2,R1     ; 2/pi x

              ADDF      0.5,R1

              FIX       R1,R0           ; integer part

              FLOAT     R0,R3

              AND       3,R0            ; R0 = remainder

              ADDI      R0,AR0,AR1

              LDI       *+AR1(2),R0

              BUD       R0

              SUBRF     R1,R3           ;\

              SUBF      0.5,R3          ; > delay

              MPYF      *+AR0,R3,R2     ;/     fractional

;   <<<       jump from here

quart0        LDI       @sin20data,AR0  ; constant!!! (ARx1)

              MPYF      R2,R2,R3        ; x^2

              MPYF      *AR0,R3,R0

              ADDF      *+AR0,R0

              MPYF      R3,R0

              ADDF      *+AR0(2),R0

              MPYF      R2,R0

              RETS

quart1        LDI       @cos16data,AR0  ; constant!!! (ARx2)

              MPYF      R2,R2,R3

              MPYF      *AR0,R3,R0

              ADDF      *+AR0,R0

              MPYF      R3,R0

              ADDF      *+AR0(2),R0

              RETS

quart2        LDI       @nsin20data,AR0 ; constant!!! (ARx3)

              MPYF      R2,R2,R3

              MPYF      *AR0,R3,R0

              ADDF      *+AR0,R0

              MPYF      R3,R0

              ADDF      *+AR0(2),R0

              MPYF      R2,R0

              RETS

quart3        LDI       @ncos16data,AR0 ; constant!!! (ARx4)

              MPYF      R2,R2,R3

              MPYF      *AR0,R3,R0

              ADDF      *+AR0,R0

              MPYF      R3,R0

              ADDF      *+AR0(2),R0

              RETS

The above method for finding subintervals where the approximation is easy is very useful in general. Here 6 DSP instructions are needed to get the jump address from a table, a delayed jump (with useful instructions in the pipeline) needs only one extra step.

atan(y/x) for small angles

The atan(y/x) is the most often used form of the inverse tangent. It is needed to find the angle of a vector given by its Cartesian coordinates (x,y). We cannot avoid one (slow) division, but if we must perform one, why not try a rational approximation instead of a polynomial one? We can still get away with one division only. For example:

. With coefficient optimization we found the simplest rational approximation in [–( /8, ( /8]
. It is of 14-bit accuracy:

This accuracy and the length of the approximation interval are large enough for many applications. If not, we have to change the approximation formula (see the section Weighted Least-Squares Optimization below) or cut the interval into smaller pieces and rotate the vector to get it in the approximation interval (see the next 2 sections).

; float atan14(float Y,float X)
 =atan2(Y,X)
 err +/-5.62e-5 >14bit if |angle(X,Y)|<pi/8

; 15 DSP instructions + Initialization and CALL/RETURN

atan14data   .word      atan14data+2    ; allocate in the best RAM block

             .word      0FF7FFFFFh      ; *-AR0

             .word      0               ; *AR0 Middle of constant block

             .float     0.786837        ; *+AR0

             .float     2.253905        ; *+AR0(2)

             .float     0.315848        ; *+AR0(3)

_atan14       LDI       @atan14data,AR0 ; constant!!!

;;; CALCULATE NUMERATOR, DENOMINATOR

              MPYF      R2,R2,R0        ; Y^2

              MPYF      *+AR0(3),R0     ; 0.32 Y^2

              MPYF      R3,R3,R1        ; X^2

              MPYF      R2,R3           ; X <- XY

              ADDF      R0,R1,R2        ; Y <- denominator

              STF       R2,*AR0         ; store denominator

;;; RECIPROCAL in 19 bit accuracy

              XOR       *AR0,*-AR0,R0   ; flip all but sign bit

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      R2,*AR0,R1      ; xy

              SUBRF     2.125,R1        ; 2.125-xy [2.125 = 2/0.9412: symmetric error]

              MPYF      R1,*AR0++,R0    ; y*(2.125-xy), move pnter for parallel addrsng

              MPYF      R2,R0,R1        ; xy (R0=y,R1=xy,R2=x))

              MPYF      *AR0--,R0       ; scale y, restore pntr (cycle lost)

           || SUBF      R1,*+AR0,R2     ; 2.2539-xy [2/0.88735: symmetric error]

              MPYF      R2,R0           ; y*(2.2539-xy)

              MPYF      R3,R0           ; XY / (X^2 + 0.32 Y^2)

              RETS

Octant of vector angle

For the general inverse tangent, among other applications, we need to find the approximate direction of a vector. To locate the quadrant of the coordinate system where a vector lies is easy: look at the signs of the coordinates. (If both x,y > 0, it is quadrant 0, if x < 0, y > 0, it is quadrant 1, etc.) With a little extra work we can also tell if the vector points to the upper or lower half a quadrant, that is to which octant it belongs.

The usual way is a series of compare-branch instructions. In a pipelined processor this is slow, because a conditional jump often flushes the pipeline. It is better to collect the information and do one single branch at the end. During the collection process we construct a value that will be used as an index in a jump table. Each bit of this value represents the result of a comparison as follows. A shift-right operation moves the sign bit of a number to the least significant bit of a register R. Later, when comparing 2 numbers we store their difference in another register, shift it left, such that its sign bit gets to the carry-bit storage (in the status register of the DSP). A rotate-left-with-carry instruction appends this saved sign bit to the bit sequence in R.

; int octant(float y, float x) = 
octant number

;     keeps xy, x^2, y^2, x^2-y^2 for the subsequent atng()

; Except C administration, constant init: octant = 8 inst; jump address = 3 inst


_octant:      PUSH      R4              ; octant number

              PUSH      R5              ; temp results

              LDI       @JumpTable-2,AR0; constant!!!

              MPYF      R3,R2,R1        ; R1 = xy (only used in atng)

              LSH       *+AR0,R3,R4     ; sign(x) -> R4

              LSH       *-AR0,R2,R5     ; sign(y) -> C

              ROLC      R4              ; R4 <<- C

              MPYF      R3,R3           ; R3 = x^2

              MPYF      R2,R2           ; R2 = y^2

              SUBF      R2,R3,R0        ; R0 = x^2-y^2

              LSH       *-AR0,R0,R5     ; sign -> C

              ROLC      R4              ; R0 <<- C

              ADDI      R4,AR0,AR1      ; table + offset

              LDI       *+AR1(2),R4     ; jump destination

              BUD       R4

              NOP                       ;\  <- LDI @atng14data,AR0

              POP       R5              ; > delay

              POP       R4              ;/

;   <<<       jump from here

             .word      1               ; *-AR0

             .word      JumpTable-2     ; *AR0 Middle of constant block

             .word      -31             ; *+AR0

JumpTable    .word      octant0         ; 0:      + atng14(y,x)

             .word      octant1         ; 1: pi/2 - atng14(x,y)

             .word      octant7         ; 2: pi/2 + atng14(-x,y)

             .word      octant6         ; 3:  pi  - atng14(y,-x)

             .word      octant3         ; 4: -pi  + atng14(-y,-x)

             .word      octant2         ; 5:-pi/2 - atng14(-x,-y)

             .word      octant4         ; 6:-pi/2 + atng14(x,-y)

             .word      octant5         ; 7:      - atng14(-y,x)

octant0       LDI       0,R0
            ;  
              RETS

octant1       ...

1st octant atan(y/x)

Once we know the coordinate octant of a vector, we can reduce the inverse tangent calculation to the 1st octant, where: x, y > 0, x > y. It means, the angle of the vector (x,y) is between 0 and ( /4
, different from the range where we have fast and accurate approximations. If we reuse the partial results of the octant determination only 2 more DSP instructions are required for the atan calculation than in the range [–( /8, ( /8]
.
The rotation of the complex number x+i·y
 by the angle –( /8
 is the same as multiplying it by 

. Denoting the result by 

. After doing the math we get the approximation formula of max error: 

  if x ( y ( 0, x > 0.

; float atng14(float Y,float X) =atan2(Y,X) err +/-5.6e-5 > 14bit if 0<angle(X,Y)<pi/4




; 21 DSP instructions + Initialization and CALL/RETURN

atng14data   .word      atng14data+2    ; allocate in the best RAM block

             .word      0FF7FFFFFh      ; *-AR0

             .word      0               ; *AR0 Middle of constant block

             .float     0.57504557      ; *+AR0    (1/x correction)/b3

             .float     2.253905        ; *+AR0(2)

             .float     1.85999755      ; *+AR0(3)  b1/b3

             .float     0.85999755      ; *+AR0(4)  b2/b3 (b1/b3-1))

             .float     0.39269908      ; *+AR0(5)  pi/8

_atng14       LDI       @atng14data,AR0 ; constant!!!

;;; CALCULATE NUMERATOR, DENOMINATOR

              MPYF      R2,R3,R1        ; XY

              MPYF      R2,R2           ; Y^2

              MPYF      R3,R3           ; X^2

              SUBF      R3,R2,R0        ; Y^2 - X^2

* >> Start here for 4 quadrant atan2

              MPYF      *+AR0(3),R3     ; b1/b3 X^2

              MPYF      *+AR0(4),R2     ; b2/b3 Y^2

              ADDF      R3,R2           ; sum

              ADDF      R1,R0

              ADDF      R1,R0,R3        ; R3 = Y^2 - X^2 + 2 XY

              ADDF      R1,R2           ; R2 = denominator

              STF       R2,*AR0         ; store denominator to RAM

;;; RECIPROCAL in 19 bit accuracy 1/R2

              XOR       *AR0,*-AR0,R0   ; flip all but sign bit

              STI       R0,*AR0         ; write 32 bits to RAM

              MPYF      R2,*AR0,R1      ; xy

              SUBRF     2.125,R1        ; 2.125-xy [2.125 = 2/0.9412: symmetric error]

              MPYF      R1,*AR0++,R0    ; y*(2.125-xy), move pnter for parallel addrsng

              MPYF      R2,R0,R1        ; xy (R0=y,R1=xy,R2=x))

              MPYF      *AR0--,R0       ; scale y, restore pointer (cycle lost)

           || SUBF      R1,*+AR0,R2     ; 2.2539-xy [2/0.88735: symmetric error]

              MPYF      R2,R0           ; y*(2.2539-xy)

              MPYF      R3,R0           ; XY / (X^2 + 0.32 Y^2)

              ADDF      *+AR0(5),R0     ; + pi/8

              RETS
Putting the pieces together gives a 14 bit accurate atan(y/x) in 32 DSP cycles.
median-of-3 filter

As already discussed, the compare-branch approach is too slow in pipelined processors. Modern DSPs have instructions that can be performed conditionally. The C3x chips only have conditional load instructions, but these are powerful enough to implement algorithms traditionally performed using branches. One good example is the non-linear, median-of-3 filter. It takes every 3 consecutive members of a sequence and returns for each triplet the median, the middle one of the sorted 3-sequence. This filter applied to noisy data has a similar low-pass smoothing effect of a 10-tap FIR filter with smaller delay.
The algorithm is simple. For each triplet calculate all 3 differences. If 

 is the median. Load it conditionally to a register. If 

 is the median.

; void MEDI3filt( float *Data, float *Result, int DataLen);

; Result length = DataLen - 2

; 10 DSP instructions / iteration

_MEDI3filt    SUBI      3,R3

              LDI       R3,RC            ; Repeat Count for the loop

              LDI       R2,AR1           ; Address of output

              RPTB      m_End

              SUBF      *+AR2,*AR2++,R1  ; x(1) - x(2) -> R1

              SUBF      *+AR2,*AR2,R2    ; x(2) - x(3) -> R2

              ADDF      R1,R2,R3         ;-x(3) + x(1) -> R3

              XOR       R1,R2,R0         ; equal sign?

              LDFNN     *AR2,R0          ; if same: median = x(2)

              XOR       R3,R2            ; R2 not needed more

              LDFN      *+AR2,R0         ; median = x(3)

              XOR       R3,R1

              LDFN      *-AR2,R0         ; median = x(1)

m_End         STF       R0,*AR1++        ; save median

              RETS

Finding approximation polynomials
The best approximation polynomials, according to the Chebishev theory have equally large alternating sign peak errors. For a polynomial of degree n the number of these peaks is at least n+2. The well known Remez exchange algorithm and its variants (see [2]) find these polynomial approximations. However, there are practical difficulties: a good initial approximation is needed – which is not always easy to find; if at some stage of the process the number of peak errors differ from n+2, the algorithm breaks down, etc.

We used instead a weighted least-squares optimization procedure for finding the approximation polynomials listed earlier. It is more robust and the approximation it provides is so close to the best possible that only a fraction of a bit accuracy is lost.

We want to approximate a function (given as a list y of its values) by a linear combination of other functions. They are given in the columns of a matrix x as lists of values at the same arguments as at y. (At polynomial approximations the columns of x are some powers of the vector of the function argument, but other basis functions can be used, too.)

With least-squares approximations the optimum linear combination is defined as

The advantage of this kind of optimality criterion is that taking the partial derivatives of F with respect of every 

 gives linear expressions. At the optimum the derivatives are 0, which gives a linear sequence of equations. We can even specify 

 weights at each point 

 telling how important is a good approximation there. This way we can control the maximum error of the approximation.

To solve the system of equations a single MATLAB line is enough. Our 3-line code tells more what happens under the hood.

function a = curvfit(y,x,w)

y = y.* w';

x = x.* w(:,ones(1,m));

a =(x'* x)\(y*x)';

We want an approximation, where the maximum error is minimized. The approximation error can be relative to the function value, or, if further processing happens with the approximation, the final deviation from the desired. The following MATLAB program tries to achieve a mini-max approximation for the case of 

 discussed earlier.

First we run the curve fit procedure un-weighted. Calculate the approximation error at each point. Search for intervals where the errors are of the same sign. In each of these intervals find the error of the largest absolute value. Use these peaks as weights in every point in their corresponding interval: in the neighborhood where the error is large a new least-squares approximation fits better, making the maximum deviation smaller.

Repeat the same procedure a few times. In our cases 3 least-squares optimizations proved to be enough. However, there is no guarantee that this process always works. A better alternative is to let the weights accumulate at each point such that older weights contribute less and less:

The following MATALB code implements the simpler variant:

n = 500;

t = linspace(1,2,n)';

z = 2.^t;

y = z ./ (2*t);

w = ones(size(t));

x =[w,t,t.^2,t.^3,t.^4,t.^5];

a = curvfit(y,x,w); % #1 least squares approximation

e = (x*a).*(2*t)./ z - 1;

i0= 1;

e(n+1) = -e(n);

p = 1.1*e(1);       % first peak corrected

for i = 2 : n

    if (e(i0) > 0) == (e(i) > 0)

       if (e(i-1) >= e(i)) == (e(i+1) >= e(i))

          p = e(i);

       end

    else

       w([i0:i-1]) = p;

       i0 = i;

    end

end                 % last peak corrected

w([i0:n]) = 1.1*p;

w = w / max(w);     % set up new weight

b = curvfit(y,x,w); % #2 least squares

f = (x*b).*(2*t)./ z - 1;

v = ones(size(t));

i0= 1;

f(n+1) = -f(n);

p = f(1);

for i = 2 : n

    if (f(i0) > 0) == (f(i) > 0)

       if (f(i-1) >= f(i)) == (f(i+1) >= f(i))

          p = f(i);

       end

    else

       v([i0:i-1]) = p;

       i0 = i;

    end

end

v([i0:n]) = p;

v = v / max(v);     % correction for the weights

c = curvfit(y,x,w.*v);    % #3 least squares

g = (x*c).*(2*t)./ z - 1; % final apprx errors

